The Calculus of Vector-Valued Functions

```
> restart:with(plots):with(VectorCalculus):
  setoptions3d(axes=NORMAL,labels=[x,y,z],orientation=[20,70]):
  BasisFormat(false):
```

We will use the VectorCalculus package for this worksheet. We first find $\lim_{t \to \frac{\pi}{4}} \langle \sin(t)\cos(t), \cos(t)^2, \sin(t) \rangle$.

```
> limit(<sin(t)*cos(t),(cos(t))^2,sin(t)>,t=Pi/4);
```

$$\begin{bmatrix} \frac{1}{2} \\ \frac{1}{2} \\ \frac{1}{2}\sqrt{2} \end{bmatrix}$$

Next we find the derivative of **r**(t)=$\langle \sin(t)\cos(t), \cos(t)^2, \sin(t) \rangle$.

```
> r:=<sin(t)*cos(t),(cos(t))^2,sin(t)>;
```

$$r := \begin{bmatrix} \sin(t)\cos(t) \\ \cos(t)^2 \\ \sin(t) \end{bmatrix}$$

```
> rprime:=diff(r,t);
```

$$rprime := \begin{bmatrix} \cos(t)^2 - \sin(t)^2 \\ -2\sin(t)\cos(t) \\ \cos(t) \end{bmatrix}$$

We see that this derivative is never $\langle 0, 0, 0 \rangle$, since if the third component is 0, the first component is -1, so **r** is a smooth curve. We plot **r**.

```
> spacecurve(r,t=0..2*Pi);
```



On the other hand, the curve **s**(t)=$\left\langle t^3 - t^2, \frac{t^2}{2} - \frac{2\,t}{3}, \cos\left(t - \frac{2}{3}\right)\right\rangle$ is not smooth at $t = \frac{2}{3}$ since **s**'($\frac{2}{3}$)=⟨0, 0, 0⟩.

```
> s:=<t^3-t^2,t^2/2-2/3*t,cos(t-2/3)>;
```

$$s := \begin{bmatrix} t^3 - t^2 \\ \frac{1}{2}\,t^2 - \frac{2}{3}\,t \\ \cos\left(t - \frac{2}{3}\right) \end{bmatrix}$$

```
> sprime:=diff(s,t);
```

$$sprime := \begin{bmatrix} 3\,t^2 - 2\,t \\ t - \frac{2}{3} \\ -\sin\left(t - \frac{2}{3}\right) \end{bmatrix}$$

```
> spacecurve(s,t=-2..2);
```



Using the above vector functions **r**(t) and **s**(t), we illustrate that $\frac{\partial}{\partial t}\left(\boldsymbol{r}(t) \times \boldsymbol{s}(t)\right) = \frac{d}{dt}\boldsymbol{r}(t) \times \boldsymbol{s}(t) + \boldsymbol{r}(t) \times \frac{d}{dt}\boldsymbol{s}(t)$.

```
> diff(CrossProduct(r,s),t);
```

$$\left[\left[-2\sin(t)\cos(t)\cos\left(t-\frac{2}{3}\right) - \cos(t)^2\sin\left(t-\frac{2}{3}\right) - \cos(t)\left(\frac{1}{2}t^2 - \frac{2}{3}t\right) - \sin(t)\left(t - \frac{2}{3}\right)\right],\right.$$

$$\left[\cos(t)\left(t^3 - t^2\right) + \sin(t)\left(3\,t^2 - 2\,t\right) - \cos(t)^2\cos\left(t-\frac{2}{3}\right) + \sin(t)^2\cos\left(t-\frac{2}{3}\right)\right.$$

$$+\sin(t)\cos(t)\sin\left(t-\frac{2}{3}\right)\Big],$$

$$\Big[\cos(t)^2\left(\frac{1}{2}t^2-\frac{2}{3}t\right)-\sin(t)^2\left(\frac{1}{2}t^2-\frac{2}{3}t\right)+\sin(t)\cos(t)\left(t-\frac{2}{3}\right)+2\cos(t)\left(t^3-t^2\right)\sin(t)-\cos(t)^2\left(3t^2-2t\right)\Big]\Big]$$

```
> CrossProduct(diff(r,t),s)+CrossProduct(r,diff(s,t));
```

$$\Big[\Big[-2\sin(t)\cos(t)\cos\left(t-\frac{2}{3}\right)-\cos(t)^2\sin\left(t-\frac{2}{3}\right)-\cos(t)\left(\frac{1}{2}t^2-\frac{2}{3}t\right)-\sin(t)\left(t-\frac{2}{3}\right)\Big],$$

$$\Big[\cos(t)\left(t^3-t^2\right)-\left(\cos(t)^2-\sin(t)^2\right)\cos\left(t-\frac{2}{3}\right)+\sin(t)\left(3t^2-2t\right)+\sin(t)\cos(t)\sin\left(t-\frac{2}{3}\right)\Big],$$

$$\Big[\left(\cos(t)^2-\sin(t)^2\right)\left(\frac{1}{2}t^2-\frac{2}{3}t\right)+2\cos(t)\left(t^3-t^2\right)\sin(t)+\sin(t)\cos(t)\left(t-\frac{2}{3}\right)-\cos(t)^2\left(3t^2-2t\right)\Big]\Big]$$

We now look at an example of uniform circular motion with **r**(t)=$\langle 2\cos(3t), 2\sin(3t)\rangle$.

```
> r:=<2*cos(3*t),2*sin(3*t)>;
```

$$r := \begin{bmatrix} 2\cos(3t) \\ 2\sin(3t) \end{bmatrix}$$

We find the derivative.

```
> rprime:=diff(r,t);
```

$$rprime := \begin{bmatrix} -6\sin(3t) \\ 6\cos(3t) \end{bmatrix}$$

We plot the curve along with the position and tangent vectors for $t=\frac{\pi}{9}$.

```
> p1:=plot([r[1],r[2],t=0..2/3*Pi]):
  p2:=PlotVector(eval(r,t=Pi/9),width=.1,head_width=.2,color=cyan):
  v:=RootedVector(root=[eval(r[1],t=Pi/9),eval(r[2],t=Pi/9)],[eval
  (rprime[1],t=Pi/9),eval(rprime[2],t=Pi/9)]):
  p3:=PlotVector(eval(v,t=Pi/9),width=.1,head_width=.2,color=magenta)
  :
  display(p1,p2,p3,scaling=constrained);
```



The position and tangent vectors appear to be **orthogonal** for $t = \frac{\pi}{9}$.

```
> DotProduct(eval(r,t=Pi/9),eval(rprime,t=Pi/9));
```

$$0$$

The dot product verifies the orthogonality. Next, we find $\int \langle e^{-3t}, t^2 \cos(t^3), t\cos(t)\rangle \, dt$.

```
> Int(<exp(-3*t),t^2*cos(t^3),t*cos(t)>,t)=int(<exp(-3*t),t^2*cos
  (t^3),t*cos(t)>,t);
```

$$\int \begin{bmatrix} e^{-3t} \\ t^2 \cos(t^3) \\ t\cos(t) \end{bmatrix} dt = \begin{bmatrix} -\frac{1}{3} e^{-3t} \\ \frac{1}{3}\sin(t^3) \\ \cos(t) + t\sin(t) \end{bmatrix}$$

Finally, we find $\int_0^{\frac{\pi}{3}} \langle \sec(t)\tan(t), \tan(t), 2\sin(t)\cos(t)\rangle \, dt$.

```
> Int(<sec(t)*tan(t),tan(t),2*sin(t)*cos(t)>,t=0..Pi/3)=int(<sec(t)*
```

```
tan(t),tan(t),2*sin(t)*cos(t)>,t=0..Pi/3);
```

$$\int_{0}^{\frac{1}{3}\pi} \begin{bmatrix} \sec(t)\tan(t) \\ \tan(t) \\ 2\sin(t)\cos(t) \end{bmatrix} \mathrm{d}t = \begin{bmatrix} 1 \\ \ln(2) \\ \frac{3}{4} \end{bmatrix}$$